Slide 1	**Absorption Models Using NM-TRAN** Nick Holford Dept Pharmacology & Clinical Pharmacology University of Auckland, New Zealand	
Slide 2	Absorption • Extent of Absorption • Rate of Absorption 	
Slide 3	Extent (F) • Fraction Absorbed (f) – into portal vein from gut – physicochemistry • theophylline (100%) (small,unionized) • gentamicin (< 5%) (large, ionized) – metabolism/transport • simvastatin (50%?) (CYP3A4) • digoxin (65%) (PGP transporter) 	The extent of oral absorption can be considered in 2 parts. The first part is the fraction of drug absorbed across the gut wall (f). This describes how much drug gets from the gut into the portal venous system. It is determined in part by physicochemical properties. Small, unionized molecules e.g. theophylline, are almost completely absorbed across the gut wall. Large, ionized molecules like gentamicin cross membranes with difficulty and only a small fraction is absorbed across the gut wall. Many drugs are cross the luminal cell membrane but are then metabolized in the gut wall (typically by CYP3A4 e.g. simvastatin) and/or transported out of the cell back into the gut lumen (typically by P-glyco-protein e.g. digoxin).

Slide 4	**Extent (F)** • First Pass Extraction (ER) – drug removed while passing through liver – organ clearance and blood flow • morphine (60%) • ethanol (10-70%) 	
Slide 5	**Extent (F)** $F = f \cdot (1 - ER)$ e.g. morphine $F = 1 \cdot (1 - 0.6) = 0.4$ 	
Slide 6	**Input Processes** • Bolus e.g. Intra-venous injection • Zero-Order e.g. Constant rate IV infusion • First-Order e.g. Intra-muscular injection 	

Slide 7

Rate

- Zero-Order
  - Stomach emptying (Tmax)
    - physiological control
  - Slow Release Formulation (Tk0)
    - pharmaceutical control

$$C(t) = \frac{Rate}{CL} \bullet \left(1 - e^{-\frac{CL}{Vd} \bullet t}\right)$$

Slide 8

Rate

- First Order
  - Intestinal Absorption (KA)
    - Diffusion limited
    - Absorption Thalf = 0.7/KA
    - Complete after 4 x absorption Thalf

$$C(t) = \frac{Dose \bullet Ka}{V \bullet \left(Ka - \frac{CL}{Vd}\right)} \bullet \left(e^{-\frac{CL}{Vd} \bullet t} - e^{-Ka \bullet t}\right)$$

Slide 9

Extent of Absorption
Dose or Rate Dependence?

- Empirical Models use Dose
- Mechanistic Models use Rate

Slide 10	**Finding the Dose**
	
	`$PK`
	
	`;Make sure dose is 0 for each subject`
	`IF (NEWIND.LE.1) DOSE=0`
	
	`;Remember dose using AMT`
	`IF (AMT.GT.0) DOSE=AMT`
	
		NONMEM data files require that the amount of each dose is recorded in the AMT data item only at the time the dose is administered. It is often useful to know the value of the last DOSE. This code shows how to save the DOSE by looking at the AMT data item.

Slide 11	**Finding the Time After Dose**
	
	`$PK`
	`;Make sure values are 0 for each subject`
	`IF (NEWIND.LE.1) THEN`
	`  DOSE=0`
	`  TDOS=0`
	`ENDIF`
	`;Remember dose and time of dose`
	`IF (AMT.GT.0) THEN`
	`  DOSE=AMT`
	`  TDOS=TIME`
	`ENDIF`
	`;Time after dose for every record`
	`TAD=TIME-TDOS`
	
	See also
	http://www.globomax.com/nonmem_tip3.htm
	http://www.globomax.com/nonmem_tip4.htm	

Slide 12	**Empirical Dose and Extent Models**
	
	`$THETA -.01 ; KFDOSE`
	`$PK`
	
	`FDOSE=EXP(KFDOSE*DOSE)`
	
	`F1=FDOSE`
	
	`$THETA (-1,-.5,0) ; DMAX`
	`$THETA (0,100,)  ; D50`
	
	`$PK`
	
	`FDOSE=1+DMAX*DOSE/(D50+DOSE)`
	
	`F1=FDOSE`
	
		Two empirical models are illustrated showing how dose might be used to predict changes in extent of absorption. The model on the right is approximately linear if the effect on bioavailability is small. The model on the left requires an additional parameter to describe a saturable increase in extent of absorption with dose. The maximum extent is 1 + DMAX with 1+DMAX/2 extent at a dose D50.

Slide 13

Mechanistic Rate and Extent
Well Stirred Model

```
$DES
;Systemic circulation conc
DCP=A(1)/V
;Rate in to Portal vein
RATEIN=KA*A(2)
;Portal vein conc
CPV=DCP+RATEIN/Q

;Solve quadratic for CLI
AXX  =KM
B    =Q*(CPV+KM)-VMAX
C    =-VMAX*Q
SGNB=1
IF (B.LT.0) SGNB=-1
SQRB =SQRT(B*B-4*AXX*C)
D    =-.5*(B+SGNB*SQRB)
ISPLUS=1
IF (C/D.LE.0) ISPLUS=0
CLI  =ISPLUS*C/D+(1-ISPLUS)*D/AXX

ER   =CLI/(Q+CLI)

; mixed order component of CL
CLMO = Q*ER
FHEP = 1-ER
CL   = CLMO + CLFO

;CP
DADT(1)= FHEP*RATEIN - CL*DCP

;GUT
DADT(2)= -RATEIN
```


Hepatic extraction can change with concentration delivered to the liver if intrinsic clearance is mixed order in the range of concentrations achieved in the liver during absorption. Under the well stirred model of liver extraction the time changing extraction ratio (ER) can be predicted from the absorption rate and an assume value for hepatic blood flow. CLFO is a parameter describing a first-order elimination clearance process.

http://pkpdrx.com/holford/docs/rate-dependent-extraction.pdf

Slide 14

Two Part Rate Models
Parallel/Sequential K0KA

Dose is 100

```
#ID TIME CMT AMT  RATE
1    0    1   100  0
1    0    2   100 -2
```

```
$THETA
1   ; ka h-1
.25 ; tlag1 h
.1  ; tlag2 h
2   ; tk0 h
.5  ; fk0

$SUBR ADVAN2
$PK
KA=ka
D2=tk0
F1=1-fk0
F2=fk0
ALAG1=tlag1
ALAG2=tlag2 ;+tlag1
```


Two part models require the data file to have two dosing records at the time of each actual dose. The first record has a CMT of 1 (the default absorption compartment for a first-order process when using ADVAN2 or ADVAN4) and a RATE of 0. The second record has a CMT of 2 (the default central compartment when using ADVAN2 or ADVAN4) and a RATE of -2. The duration of input (TK0) by the zero-order process into compartment 2 is defined in $PK by assigning TK0 to the special variable D2. The fraction of the dose absorbed by each process is determined by FK0 (the fraction absorbed by the zero-order process).
ALAG1 and ALAG2 are used to estimate the lagtime (if any) for each process. If it is assumed that the zero-order process starts only when the first-order process has started (sequential input) then ALAG2 must be set to the sum of tlag2 + tlag1.

Slide 15

Two Part Rate Models
Sequential Linked K0KA

When Zero-order becomes First-order
Rate Zero-order = Rate First-order i.e.

```
K0 = KA*GUT
   = KA*(1-FK0)*DOSE
FK0*DOSE/TK0 = KA*(1-FK0)*DOSE
```

therefore

```
KA = FK0/(TK0*(1-FK0))
```

```
#ID TIME CMT AMT  RATE
1    0    1   100  0
1    0    2   100 -2
```

```
$THETA
.25 ; tlag h
2   ; tk0 h
.5  ; fk0

$SUBR ADVAN2
$PK
KA=fk0/(tk0*(1-fk0))
D2=tk0
F1=1-fk0
F2=fk0
ALAG1=tlag+tk0
ALAG2=tlag
```


The sequential linked zero-order then first-order model can be interpreted in a mechanistic way if the concentration in the gut is initially at the solubility limit for the drug. As drug is absorbed and the concentration falls below the solubility limit then the process converts from a zero-order input to a first-order input. Note that ALAG1 (lag time of the first order process) must be equal to TK0 (the duration of the zero-order process) plus any lagtime for the zero-order process. This ensures that the first-order process will take over at the end of the zero-order input.
See Holford NHG, Ambros R, Stoeckel K. Models for describing absorption rate and estimating extent of bioavailability: Application to cefetemet pivoxil. J Pharmacokin Biopharm. 1992;20:421-42.

Slide 16	# Transit Models UPPSALA UNIVERSITET *Evaluation of a transit compartment model versus a lag time model for describing drug absorption delay* *Radojka Savic(1), Daniël M. Jonker(1), Thomas Kerbusch(2), Mats O. Karlsson(1)* *(1) Div. of Pharmacokinetics and Drug Therapy, Dept of Pharmaceutical Biosciences, Faculty of Pharmacy, Uppsala University, Sweden.* *(2) Dept. Clinical PK/PD M&S, Pfizer Global R&D, Sandwich, Kent, UK*	The use of transit compartment models to describe delays in onset of absorption was developed by Radojka Savic at Uppsala. The following set of slides is taken from her work.
Slide 17	# Transit Models UPPSALA UNIVERSITET *Transit compartment model* 	
Slide 18	# Transit Models UPPSALA UNIVERSITET *Step-wise addition* 	

Slide 19		
Slide 20		
Slide 21		

Slide 22	**Transit Models**
	Single Dose
	
	`$THETA`
	`3 ; pop_cl L/h`
	`10 ; pop_v L`
	`1 ; pop_ka h-1`
	`1 ; pop_mtt h`
	`5 ; pop_nt`
	
	`$SUBR ADVAN6 TOL=3`
	`$MODEL`
	`COMP (GUT)`
	`COMP (CENTRAL)`
	
	`$PK`
	`CL=pop_cl`
	`V =pop_v`
	`KA=pop_ka`
	`NT=pop_nt`
	`KTR=(NT+1)/pop_mtt`
	`NFAC= SQRT(2*3.1415)*NT**(NT+0.5)*EXP(-NT)`
	`F1=0 ; Very important!`
	
	`$DES`
	`DCP=A(2)/V`
	`RATEIN=KA*A(1)`
	`GUT=DOSE*EXP(-KTR*(T-TLAST))`
	`DADT(1)=GUT*KTR*(KTR*(T-TLAST))**NT/NFAC - RATEIN`
	`DADT(2)=RATEIN - CL*DCP`
	
	©NHG Holford, 2019, all rights reserved.
		The code for a single dose transit model involves the use of differential equations and closed form solution to transit through NT transit compartments. It is important that the extent of absorption for the dosing compartment 1 is set to 0 because the actual drug input rate from this compartment is modelled explicitly in DADT(1).

Slide 23	**Transit Models**
	Single Dose LOG Version (faster?)
	
	`$THETA`
	`3 ; pop_cl L/h`
	`10 ; pop_v L`
	`1 ; pop_ka h-1`
	`1 ; pop_mtt h`
	`5 ; pop_nt`
	
	`$SUBR ADVAN6 TOL=3`
	`$MODEL`
	`COMP (GUT)`
	`COMP (CENTRAL)`
	
	`$PK`
	`CL=pop_cl`
	`V =pop_v`
	`KA=pop_ka`
	`NT=pop_nt`
	`KTR=(NT+1)/pop_mtt`
	`LNFAC= LOG(2.5066)+(NT+.5)*LOG(NT)-NT`
	`LNDK=LOG(DOSE+.00001)+LOG(KTR)`
	`F1=0 ; Very important!`
	
	`$DES`
	`DCP=A(2)/V`
	`RATEIN=KA*A(1)`
	`X=KTR*T`
	`DADT(1)=EXP(LNDK+NT*LOG(X+0.00001)-X-LNFAC)-RATEIN`
	`DADT(2)=RATEIN - CL*DCP`
	
	©NHG Holford, 2019, all rights reserved.	

Slide 24	**Transit Models**
	Multiple Dose
	Absorption must be complete within dosing interval!
	
	`$THETA`
	`3 ; pop_cl L/h`
	`10 ; pop_v L`
	`1 ; pop_ka h-1`
	`1 ; pop_mtt h`
	`5 ; pop_nt`
	
	`$SUBR ADVAN9 TOL=3`
	`$MODEL`
	`COMP (GUT)`
	`COMP (TRANSIT)`
	`COMP (CENTRAL)`
	
	`$PK`
	`CL=pop_cl`
	`V =pop_v`
	`KA=pop_ka`
	`NT=pop_nt`
	`KTR=(NT+1)/pop_mtt`
	`NFAC= SQRT(2*3.1415)*NT**(NT+0.5)*EXP(-NT)`
	`F1=1 ; Very important!`
	`IF (AMT.GT.0) TDOSE=TIME`
	`$DES`
	`DCP=A(3)/V`
	`GUT=A(1)`
	`RATEIN=KA*A(2)`
	`DADT(1)=-KTR*GUT`
	`DADT(2)=GUT*KTR*(KTR*(T-TLAST))**NT/NFAC - RATEIN`
	`DADT(3)=RATEIN - CL*DCP`
	`$ERROR`
	`TLAST=TDOSE`
	
	©NHG Holford, 2019, all rights reserved.
		If it can be assumed that absorption is complete within the dosing interval then it is possible to model concentrations arising from multiple doses with a transit time delay after each dose. In this case the first compartment is used to model input from the most recent dose and its extent of bioavailability must be set to 1.